
1 Corrections required

1.1 f_TC_22_4_15
	Function name
	f_TC_22_4_15 

	Reason for change
	1. The transmission of the RRC Connection Resume message as MSG4 at Step 5 on Ncell1 and on Ncell 11 at Step 9 has to be pre-configured to ensure that MSG4 is sent together with the contention resolution identity.

2. Re-establishment of AS security has to be configured at the local end after it has been released at Step 2 and at Step 7

	Summary of change
	1. In order to pre-configure the RRC Connection Resume message, several changes have to be done, all the changes in various functions are detailed below.

2. AS security is re-established at the local end after Step 2 and Step 7.

	TTCN module
	NBIOT_RRC_UserPlane.ttcn

	MCC160 Comment
	


Before Change:

	function f_TC_22_4_15_NBIOT() runs on NBIOT_PTC

  {

    var EUTRA_ASN1_ResumeIdentity_r13_Type v_ResumeIdentityStep2 := tsc_CIOT_ResumeIdentity_Def;

    var EUTRA_ASN1_ResumeIdentity_r13_Type v_ResumeIdentityStep7 := not4b v_ResumeIdentityStep2;

    var EUTRA_ASN1_ShortMAC_I_Type v_ShortMAC_I;

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT0 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell11, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT1 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell11, -73)

    };

    f_NBIOT_Init(c3, USER_PLANE);

    f_NBIOT_CellInfo_InitMaxReferencePower(nbiot_Cell11, -73);

    // Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, USER_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell11, USER_PLANE);

    f_NBIOT_Preamble(nbiot_Cell1, USER_PLANE);

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, USER_PLANE, RRC_CONNECTED);

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT0);  // @sic R5s170515 change 8.1 sic@

    f_NBIOT_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    f_UT_SwitchOnUE(UT, false);

    f_NBIOT_InitialRegistration(nbiot_Cell1, USER_PLANE, NORMAL, STATE2_NB_CONNECTED_UPDATED);         // @sic R5s170515 additional change: no call of the preamble sic@

    //@siclog "Step 2" siclog@

    f_NBIOT_RRCConnectionSuspend(nbiot_Cell1, v_ResumeIdentityStep2);

    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell1, -, tsc_VarShortMAC_ResumeDiscriminator);   // @sic R5s170515 change 8.4: tsc_VarShortMAC_ResumeDiscriminator sic@

    //@siclog "Step 3" siclog@

    f_NBIOT_UE_Page(nbiot_Cell1);

    //@siclog "Step 4 - 6" siclog@

    f_NBIOT_RRCConnectionResume(nbiot_Cell1, v_ResumeIdentityStep2, v_ShortMAC_I, mt_Access, omit);

    //@siclog "Step 7" siclog@

    f_NBIOT_RRCConnectionSuspend(nbiot_Cell1, v_ResumeIdentityStep7);

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell11);          // @sic R5s170515 change 8.3: set msg 4 to None for cell11 sic@

    //@siclog "Step 8" siclog@

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 9" siclog@

    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell1, nbiot_Cell11, tsc_VarShortMAC_ResumeDiscriminator);   // @sic R5s170515 change 8.4: tsc_VarShortMAC_ResumeDiscriminator sic@

    f_NBIOT_508CheckRRCConnectionResumeOnNewCell(nbiot_Cell11, v_ResumeIdentityStep7, v_ShortMAC_I, mo_Signalling);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, " Step 9");

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell11, USER_PLANE, N1_IDLE);    // @sic R5s170515 change 8.4: N1_IDLE instead of N2_CONNECTED sic@

  }




After Change:

	    function f_TC_22_4_15_NBIOT() runs on NBIOT_PTC

  {

    var EUTRA_ASN1_ResumeIdentity_r13_Type v_ResumeIdentityStep2 := tsc_CIOT_ResumeIdentity_Def;

    var EUTRA_ASN1_ResumeIdentity_r13_Type v_ResumeIdentityStep7 := not4b v_ResumeIdentityStep2;

    var EUTRA_ASN1_ShortMAC_I_Type v_ShortMAC_I;

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT0 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell11, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT1 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell11, -73)

    };

    var NBIOT_SecurityParams_Type v_SecurityParams; //WA#22_4_15
    f_NBIOT_Init(c3, USER_PLANE);

    f_NBIOT_CellInfo_InitMaxReferencePower(nbiot_Cell11, -73);

    // Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, USER_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell11, USER_PLANE);

    f_NBIOT_Preamble(nbiot_Cell1, USER_PLANE);

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, USER_PLANE, RRC_CONNECTED);

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT0);  // @sic R5s170515 change 8.1 sic@

    f_NBIOT_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    f_UT_SwitchOnUE(UT, false);

    f_NBIOT_InitialRegistration(nbiot_Cell1, USER_PLANE, NORMAL, STATE2_NB_CONNECTED_UPDATED);         // @sic R5s170515 additional change: no call of the preamble sic@

    //@siclog "Step 2" siclog@

    f_NBIOT_RRCConnectionSuspend(nbiot_Cell1, v_ResumeIdentityStep2, true);  //WA#22_4_15
    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell1, -, tsc_VarShortMAC_ResumeDiscriminator);   // @sic R5s170515 change 8.4: tsc_VarShortMAC_ResumeDiscriminator sic@

    // WA# added security activation

    v_SecurityParams := f_NBIOT_Security_Get();

    v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(nbiot_Cell1, v_SecurityParams, rrcConnectionResume);

    f_NBIOT_Security_Set(v_SecurityParams);
    //@siclog "Step 3" siclog@

    f_NBIOT_UE_Page(nbiot_Cell1);

    //@siclog "Step 4 - 6" siclog@

    f_NBIOT_RRCConnectionResume(nbiot_Cell1, v_ResumeIdentityStep2, v_ShortMAC_I, mt_Access, omit);

    //@siclog "Step 7" siclog@

    f_NBIOT_RRCConnectionSuspend(nbiot_Cell1, v_ResumeIdentityStep7, -, false); //WA#22_4_15
    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell11,-,-,cs_RRCConnectionResume_NB(tsc_RRC_TI_Def, cs_RadioResourceConfigDedicated_NB_DrbConfig({ cs_508_NB_DRB_ToAddMod_DEFAULT(tsc_DRB1) }), omit));          

    //@siclog "Step 8" siclog@

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 9" siclog@

    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell1, nbiot_Cell11, tsc_VarShortMAC_ResumeDiscriminator);   // @sic R5s170515 change 8.4: tsc_VarShortMAC_ResumeDiscriminator sic@

    // WA# added security activation

    v_SecurityParams := f_NBIOT_Security_Get();

    v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(nbiot_Cell11, v_SecurityParams, rrcConnectionResume);

    f_NBIOT_Security_Set(v_SecurityParams);
    f_NBIOT_508CheckRRCConnectionResumeOnNewCell(nbiot_Cell11, v_ResumeIdentityStep7, v_ShortMAC_I, mo_Signalling);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, " Step 9");

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell11, USER_PLANE, N1_IDLE, false);    // @sic R5s170515 change 8.4: N1_IDLE instead of N2_CONNECTED sic@ WA#22_4_15

  }



1.2 f_NBIOT_RRCConnectionSuspend
	Function name
	f_NBIOT_RRCConnectionSuspend

	Reason for change
	The message RRC Connection Resume in 22.4.15 and the RRC Connection Reject in 22.4.16 should be preconfigured as MSG4. Also, there is a need to distinguish between different Ncells to take care of the release of the DRB

	Summary of change
	Changes include introduction of flags to ensure either RRC Connection Resume is configured or RRC Connection Reject is configured. The flag p_SameCell controls the release of the DRB as in 22.4.15 and 22.4.16, the DRB does not get configured on Ncell11 after RRC connection resumes on this cell.

	TTCN module
	NBIOT_RRCSteps.ttcn

	MCC160 Comment
	


Before Change:

	   function f_NBIOT_RRCConnectionSuspend(NBIOT_CellId_Type p_CellId,

                                        EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := tsc_CIOT_ResumeIdentity_Def) runs on NBIOT_PTC

  {

    var template (value) DL_DCCH_Message_NB v_RRCConnectionRelease := cs_RRCConnectionRelease_NB(-, rrc_Suspend, p_ResumeIdentity);

    f_NBIOT_SS_ConfigRachProcedureMsg4(p_CellId);    /* => Msg4 of the RACH procedure is set to none i.e. no RRC message is sent together with the contention resolution.

                                                      * NOTE: in contrast to the RRCConnectionSetup the RRCConnectionResume is an SRB1 message and SRB1 shall be restored by the UE before sending the RRCConnectionResumeRequest and

                                                      *       according to 36.331 clause 5.3.3.3a when sending the RRCConnectionResumeRequest the UE restores RRC configuration, security context, PDCP and SRB1

                                                      *       => The UE should know about MAC and physical layer configuration and there should be no problem sending the RRCConnectionResume as separate message */

    f_NBIOT_RRC_ConnectionRelease_Common(p_CellId, USER_PLANE, v_RRCConnectionRelease);

/*     !!!! AS security needs to be restored: !!!! */

/*     v_SecurityParams := f_NBIOT_Security_Get(); */

/*     v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(p_CellId, v_SecurityParams, rrcConnectionResume); ??? which CellId to be used ??? */

/*     f_NBIOT_Security_Set(v_SecurityParams); */

/*     f_NBIOT_SS_ConfigRachProcedureMsg4 to be called here (for new cell ???) */

/*     !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!! */

  }
 


After Change:

	function f_NBIOT_RRCConnectionSuspend(NBIOT_CellId_Type p_CellId,

                                        EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := tsc_CIOT_ResumeIdentity_Def,

                                        boolean p_ResumeFlag := false,
                                        boolean p_SameCell := true) runs on NBIOT_PTC

  {

    var template (value) DL_DCCH_Message_NB v_RRCConnectionRelease := cs_RRCConnectionRelease_NB(-, rrc_Suspend, p_ResumeIdentity);

    //WA#22_4_15

    var EUTRA_ASN1_RRC_TransactionIdentifier_Type v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) DRB_ToAddModList_NB_r13 v_DRB_ToAddModList := { cs_508_NB_DRB_ToAddMod_DEFAULT(tsc_DRB1) };

    var template (value) RadioResourceConfigDedicated_NB_r13 v_RadioResourceConfigDedicated := cs_RadioResourceConfigDedicated_NB_DrbConfig(v_DRB_ToAddModList);

    var template (value) DL_DCCH_Message_NB v_RRCConnectionResume := cs_RRCConnectionResume_NB(v_RRC_TI, v_RadioResourceConfigDedicated, omit);
    if(p_SameCell) {

        if ( p_ResumeFlag ) {

          f_NBIOT_SS_ConfigRachProcedureMsg4(p_CellId,-,-,v_RRCConnectionResume);//WA#22_4_15
        } else {

          f_NBIOT_SS_ConfigRachProcedureMsg4(p_CellId,cs_RRCConnectionReject_NB);  //WA#22_4_16

        }

        f_NBIOT_RRC_ConnectionRelease_Common(p_CellId, USER_PLANE, v_RRCConnectionRelease);

    } else { 

     f_NBIOT_RRC_ConnectionRelease_Common(p_CellId, USER_PLANE, v_RRCConnectionRelease,-,-, false);

    }
/*     !!!! AS security needs to be restored: !!!! */

/*     v_SecurityParams := f_NBIOT_Security_Get(); */

/*     v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(p_CellId, v_SecurityParams, rrcConnectionResume); ??? which CellId to be used ??? */

/*     f_NBIOT_Security_Set(v_SecurityParams); */

/*     f_NBIOT_SS_ConfigRachProcedureMsg4 to be called here (for new cell ???) */

/*     !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!! */

  }



1.3 f_NBIOT_RRCConnectionResume
	Function name
	f_NBIOT_RRCConnectionResume

	Reason for change
	As the RRC Connection Resume is being preconfigured as MSG4, it is not necessary to have code to send the message in this function. Also, the activation of the AS security should be configured after Step 2.

	Summary of change
	The code related to sending of RRC Connection Resume is removed as well as activation of AS security.

	TTCN module
	NBIOT_RRCSteps.ttcn

	MCC160 Comment
	


Before Change:

	function f_NBIOT_RRCConnectionResume(NBIOT_CellId_Type p_CellId,

                                       template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,

                                       template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,

                                       template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?,

                                       template NAS_UL_Message_Type p_Expected_NAS_UL_Message := *,

                                       template (omit) RRCConnectionResume_NB_r13_IEs.drb_ContinueROHC_r13 p_ContinueROHC := omit) runs on NBIOT_PTC return template (omit) NAS_UL_Message_Type

  { /* NOTE: RRCConnectionResume is sent on SRB1 as after contention resolution (according to 36.508 Table 8.1.6.1-11 and Table 8.1.6.3-11 RRCConnectionResume does not contain any physical layer configuration like UESS configuration */

    var EUTRA_ASN1_RRC_TransactionIdentifier_Type v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) DRB_ToAddModList_NB_r13 v_DRB_ToAddModList := { cs_508_NB_DRB_ToAddMod_DEFAULT(tsc_DRB1) };

    var template (value) RadioResourceConfigDedicated_NB_r13 v_RadioResourceConfigDedicated := cs_RadioResourceConfigDedicated_NB_DrbConfig(v_DRB_ToAddModList);
    var template (present) SecurityHeaderType v_ExpectedSecurityStatus;

    var NBIOT_SecurityParams_Type v_SecurityParams;

    var NB_SRB_COMMON_IND v_NB_SRB_COMMON_IND;

    var NAS_UL_Message_Type v_NAS_UL_Message;

    var template (omit) NAS_UL_Message_Type v_PiggybackedNasMessage := omit;

    if (ischosen(p_Expected_NAS_UL_Message.tRACKING_AREA_UPDATE_REQUEST)) {  /* @sic R5s170515 change 3.1: acc. to 24.301 cl. 4.4.5 TRACKING AREA UPDATE REQUEST is always unciphered sic@ */

      v_ExpectedSecurityStatus := tsc_SHT_IntegrityProtected;

    } else {

      v_ExpectedSecurityStatus := tsc_SHT_IntegrityProtected_Ciphered;

    }

    f_NBIOT_ULGrantTransmission(p_CellId, cs_NB_UL_GrantScheduling_Start);   /* @sic R5s170515 change 3.2 sic@ */

    SRB.receive(car_NB_SRB0_RrcPdu_IND(p_CellId, cr_RRCConnectionResumeRequest_NB(p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause)));

    v_SecurityParams := f_NBIOT_Security_Get();

    v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(p_CellId, v_SecurityParams, rrcConnectionResume);

    f_NBIOT_Security_Set(v_SecurityParams);
    SRB.send(cas_NB_SRB_RrcPdu_REQ(p_CellId, tsc_SRB1, cs_TimingInfo_Now, cs_RRCConnectionResume_NB(v_RRC_TI, v_RadioResourceConfigDedicated, p_ContinueROHC)));
    alt {

      [] SRB.receive(car_NB_SRB_RrcNasPdu_IND(p_CellId, tsc_SRB1, cr_RRCConnectionResumeComplete_NB(v_RRC_TI), cr_NAS_Indication(v_ExpectedSecurityStatus, ?))) /* @sic R5s170772 change 2.3: v_ExpectedSecurityStatus sic@ */

        -> value v_NB_SRB_COMMON_IND {

        v_NAS_UL_Message := v_NB_SRB_COMMON_IND.Signalling.Nas[0].Pdu.Msg;

        if (not match(v_NAS_UL_Message, p_Expected_NAS_UL_Message)) {

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "RRCConnectionResumeComplete: Piggybacked NAS message does not match");

        }

        v_PiggybackedNasMessage := v_NAS_UL_Message;                         /* @sic R5s170515 change 3.3 sic@ */

      }

      [] SRB.receive(car_NB_SRB_RrcPdu_IND(p_CellId, tsc_SRB1, cr_RRCConnectionResumeComplete_NB(v_RRC_TI))) {

        if (ispresent(p_Expected_NAS_UL_Message)) {

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "RRCConnectionResumeComplete: Piggybacked NAS message expected");

        }

      }

    }

    f_NBIOT_SS_ConfigRachProcedureMsg4(p_CellId, f_NBIOT_508_RRCConnectionSetup());

    return v_PiggybackedNasMessage;

  }



After Change:

	function f_NBIOT_RRCConnectionResume(NBIOT_CellId_Type p_CellId,

                                       template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,

                                       template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,

                                       template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?,

                                       template NAS_UL_Message_Type p_Expected_NAS_UL_Message := *,

                                       template (omit) RRCConnectionResume_NB_r13_IEs.drb_ContinueROHC_r13 p_ContinueROHC := omit) runs on NBIOT_PTC return template (omit) NAS_UL_Message_Type

  { /* NOTE: RRCConnectionResume is sent on SRB1 as after contention resolution (according to 36.508 Table 8.1.6.1-11 and Table 8.1.6.3-11 RRCConnectionResume does not contain any physical layer configuration like UESS configuration */

    var EUTRA_ASN1_RRC_TransactionIdentifier_Type v_RRC_TI := tsc_RRC_TI_Def;
    var template (present) SecurityHeaderType v_ExpectedSecurityStatus;

    var NBIOT_SecurityParams_Type v_SecurityParams;

    var NB_SRB_COMMON_IND v_NB_SRB_COMMON_IND;

    var NAS_UL_Message_Type v_NAS_UL_Message;

    var template (omit) NAS_UL_Message_Type v_PiggybackedNasMessage := omit;

    if (ischosen(p_Expected_NAS_UL_Message.tRACKING_AREA_UPDATE_REQUEST)) {  /* @sic R5s170515 change 3.1: acc. to 24.301 cl. 4.4.5 TRACKING AREA UPDATE REQUEST is always unciphered sic@ */

      v_ExpectedSecurityStatus := tsc_SHT_IntegrityProtected;

    } else {

      v_ExpectedSecurityStatus := tsc_SHT_IntegrityProtected_Ciphered;

    }

    f_NBIOT_ULGrantTransmission(p_CellId, cs_NB_UL_GrantScheduling_Start);   /* @sic R5s170515 change 3.2 sic@ */

    SRB.receive(car_NB_SRB0_RrcPdu_IND(p_CellId, cr_RRCConnectionResumeRequest_NB(p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause)));

    alt {

      [] SRB.receive(car_NB_SRB_RrcNasPdu_IND(p_CellId, tsc_SRB1, cr_RRCConnectionResumeComplete_NB(v_RRC_TI), cr_NAS_Indication(v_ExpectedSecurityStatus, ?))) /* @sic R5s170772 change 2.3: v_ExpectedSecurityStatus sic@ */

        -> value v_NB_SRB_COMMON_IND {

        v_NAS_UL_Message := v_NB_SRB_COMMON_IND.Signalling.Nas[0].Pdu.Msg;

        if (not match(v_NAS_UL_Message, p_Expected_NAS_UL_Message)) {

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "RRCConnectionResumeComplete: Piggybacked NAS message does not match");

        }

        v_PiggybackedNasMessage := v_NAS_UL_Message;                         /* @sic R5s170515 change 3.3 sic@ */

      }

      [] SRB.receive(car_NB_SRB_RrcPdu_IND(p_CellId, tsc_SRB1, cr_RRCConnectionResumeComplete_NB(v_RRC_TI))) {

        if (ispresent(p_Expected_NAS_UL_Message)) {

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "RRCConnectionResumeComplete: Piggybacked NAS message expected");

        }

      }

    }

    f_NBIOT_SS_ConfigRachProcedureMsg4(p_CellId, f_NBIOT_508_RRCConnectionSetup());

    return v_PiggybackedNasMessage;

  }
  


1.4 f_NBIOT_SS_ConfigRachProcedureMsg4

	Function name
	f_NBIOT_SS_ConfigRachProcedureMsg4

	Reason for change
	RRC Connection Resume is a DL DCCH message, hence this needs to be taken care of in the preconfiguration of the message

	Summary of change
	Added code to handle the DL DCCH message

	TTCN module
	NBIOT_ConfigurationSteps.ttcn

	MCC160 Comment
	


Before Change:

	   function f_NBIOT_SS_ConfigRachProcedureMsg4(NBIOT_CellId_Type p_CellId,

                                              template (omit) DL_CCCH_Message_NB p_DL_CCCH_Message := omit,

                                              template (omit) RACH_TimingAdvance_Type p_RAR_TA := omit) runs on NBIOT_PTC

  { /* @sic R5-172058/R5-172053: p_RAR_TA sic@ */

    var octetstring v_EncodedMsg4;

    var template (value) NB_RachProcedureList_Type v_RachProcedureList;

    var template (value) ContentionResolution_ContainedRlcPdu_Type v_ContentionResolution_ContainedRlcPdu;

    var NBIOT_CellInfo_Type v_CellInfo;

    var RACH_TimingAdvance_Type v_RAR_TA;

    if (ispresent(p_DL_CCCH_Message)) {

      v_EncodedMsg4 := f_RRC_DL_CCCH_NB_Encvalue(valueof(p_DL_CCCH_Message));

      v_ContentionResolution_ContainedRlcPdu := cs_ContentionResolution_RlcPdu(v_EncodedMsg4);

    } else {

      v_ContentionResolution_ContainedRlcPdu := cs_ContentionResolution_RlcPdu_None;

    }

    if (ispresent(p_RAR_TA)) {

      v_RAR_TA := valueof(p_RAR_TA);

    } else {

      v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);

      v_RAR_TA := v_CellInfo.PhysicalParameters.RAR_TA;

    }

    v_RachProcedureList := cs_NB_RachProcedureList_Def(v_RAR_TA, v_ContentionResolution_ContainedRlcPdu);

    f_NBIOT_SS_CommonCellConfig(p_CellId, cads_NB_RachProcedure_Config_REQ(p_CellId, cs_TimingInfo_Now, -, -, v_RachProcedureList));

  }
 


After Change:

	  function f_NBIOT_SS_ConfigRachProcedureMsg4(NBIOT_CellId_Type p_CellId,

                                              template (omit) DL_CCCH_Message_NB p_DL_CCCH_Message := omit,

                                              template (omit) RACH_TimingAdvance_Type p_RAR_TA := omit,

                                              template (omit) DL_DCCH_Message_NB p_DL_DCCH_Message := omit /*WA#22_4_15*/) runs on NBIOT_PTC

  { /* @sic R5-172058/R5-172053: p_RAR_TA sic@ */

    var octetstring v_EncodedMsg4;

    var template (value) NB_RachProcedureList_Type v_RachProcedureList;

    var template (value) ContentionResolution_ContainedRlcPdu_Type v_ContentionResolution_ContainedRlcPdu;

    var NBIOT_CellInfo_Type v_CellInfo;

    var RACH_TimingAdvance_Type v_RAR_TA;

    if (ispresent(p_DL_CCCH_Message)) {

      v_EncodedMsg4 := f_RRC_DL_CCCH_NB_Encvalue(valueof(p_DL_CCCH_Message));

      v_ContentionResolution_ContainedRlcPdu := cs_ContentionResolution_RlcPdu(v_EncodedMsg4);

    } else if (ispresent(p_DL_DCCH_Message)) { //WA#22_4_15

      v_EncodedMsg4 := f_RRC_DL_DCCH_NB_Encvalue(valueof(p_DL_DCCH_Message));

      v_ContentionResolution_ContainedRlcPdu := cs_ContentionResolution_RlcPdu(v_EncodedMsg4);

    } else {

      v_ContentionResolution_ContainedRlcPdu := cs_ContentionResolution_RlcPdu_None;

    }

    if (ispresent(p_RAR_TA)) {

      v_RAR_TA := valueof(p_RAR_TA);

    } else {

      v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);

      v_RAR_TA := v_CellInfo.PhysicalParameters.RAR_TA;

    }

    v_RachProcedureList := cs_NB_RachProcedureList_Def(v_RAR_TA, v_ContentionResolution_ContainedRlcPdu);

    f_NBIOT_SS_CommonCellConfig(p_CellId, cads_NB_RachProcedure_Config_REQ(p_CellId, cs_TimingInfo_Now, -, -, v_RachProcedureList));

  }



New Function:

	  function f_RRC_DL_DCCH_NB_Encvalue(template (value) DL_DCCH_Message_NB p_DCCH_Message) return octetstring

  {

    var bitstring v_Bitstring := encvalue(p_DCCH_Message);

    var octetstring v_Octetstring := bit2oct(f_OctetAlignedBitString(v_Bitstring));

    return v_Octetstring;

  }




1.5 f_NBIOT_508CheckRRCConnectionResumeOnNewCell

	Function name
	f_NBIOT_508CheckRRCConnectionResumeOnNewCell

	Reason for change
	DRB is not getting configured on Ncell 11 after the resumption of the RRC Connection. Hence, while releasing the RRC Connection and release of the SRBs and DRB locally, care should be taken to check if a DRB is configured or not

	Summary of change
	Introduced a flag to check for the presence of DRB before it is released

	TTCN module
	NBIOT_NASSteps.ttcn

	MCC160 Comment
	


Before Change:

	  function f_NBIOT_508CheckRRCConnectionResumeOnNewCell(NBIOT_CellId_Type p_CellId,

                                                        template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,

                                                        template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,

                                                        template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?) runs on NBIOT_PTC

  {

    var NAS_UL_Message_Type v_NasMsg;

    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(p_CellId);

    var NAS_KsiValue v_KsiValue := f_NBIOT_SecurityKSIasme_Get();

    var template (present) NAS_UL_Message_Type v_ExpectedTAU_Request;

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var template (omit) AdditionalUpdateType v_AdditionalUpdateType;  /* @sic R5s170515 change 5: template (omit) sic@ */

    v_ExpectedTAU_Request := cr_508_TAU_Request(tsc_EpsUpdate_TaUpdate,

                                                v_KsiValue,

                                                cr_DRXparameter_Any ('5C'O) ifpresent,

                                                cr_AdditionalUpdateType_CIOT ifpresent);

    /* steps 1 - 3 */

    v_NasMsg := valueof(f_NBIOT_RRCConnectionResume(p_CellId, p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause, v_ExpectedTAU_Request));

    v_EPS_ContextStatus := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    v_AdditionalUpdateType := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType;

    /* steps 4 - 5 */

    f_NBIOT_TrackingAreaUpdate_Accept_Complete(p_CellId, USER_PLANE, tsc_SRB1, v_GutiParams, v_EPS_ContextStatus, v_AdditionalUpdateType);

    /* Step 6: Release RRC connection */

    f_NBIOT_RRC_ConnectionRelease(p_CellId, USER_PLANE, tsc_RRC_TI_Def);

  }



After Change:

	function f_NBIOT_508CheckRRCConnectionResumeOnNewCell(NBIOT_CellId_Type p_CellId,

                                                        template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,

                                                        template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,

                                                        template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?) runs on NBIOT_PTC

  {

    var NAS_UL_Message_Type v_NasMsg;

    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(p_CellId);

    var NAS_KsiValue v_KsiValue := f_NBIOT_SecurityKSIasme_Get();

    var template (present) NAS_UL_Message_Type v_ExpectedTAU_Request;

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var template (omit) AdditionalUpdateType v_AdditionalUpdateType;  /* @sic R5s170515 change 5: template (omit) sic@ */

    v_ExpectedTAU_Request := cr_508_TAU_Request(tsc_EpsUpdate_TaUpdate,

                                                v_KsiValue,

                                                cr_DRXparameter_Any ('5C'O) ifpresent,

                                                cr_AdditionalUpdateType_CIOT ifpresent);

    /* steps 1 - 3 */

    v_NasMsg := valueof(f_NBIOT_RRCConnectionResume(p_CellId, p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause, v_ExpectedTAU_Request));

    v_EPS_ContextStatus := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    v_AdditionalUpdateType := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType;

    /* steps 4 - 5 */

    f_NBIOT_TrackingAreaUpdate_Accept_Complete(p_CellId, USER_PLANE, tsc_SRB1, v_GutiParams, v_EPS_ContextStatus, v_AdditionalUpdateType);

    /* Step 6: Release RRC connection */

    f_NBIOT_RRC_ConnectionRelease(p_CellId, USER_PLANE, tsc_RRC_TI_Def,-,-,-,false); //WA#22_4_15

  }    



1.6 f_NBIOT_SS_SRBs_DRBs_Release
	Function name
	f_NBIOT_SS_SRBs_DRBs_Release

	Reason for change
	DRB is not getting configured on Ncell 11 after the resumption of the RRC Connection. Hence, while releasing the RRC Connection and release of the SRBs and DRB locally, care should be taken to check if a DRB is configured or not

	Summary of change
	Introduced a flag to check for the presence of DRB before it is released. This change affects the following functions:
1. f_NBIOT_Postamble

2. f_NBIOT_UE_Detach_SwitchOffUe
3. f_NBIOT_RRC_ConnectionRelease

4. f_NBIOT_RRC_ConnectionRelease_Common

5. f_NBIOT_RRC_ConnectionRelease_Local

6. f_NBIOT_TrackingAreaUpdate

	TTCN module
	NBIOT_ConfigurationSteps.ttcn

	MCC160 Comment
	


Before Change:

	  function f_NBIOT_SS_SRBs_DRBs_Release(NBIOT_CellId_Type p_CellId,

                                        IOT_STATE_Type p_IOT_State,

                                        template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                        DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def,

                                        template (omit) boolean p_CnfFlag := omit) runs on NBIOT_PTC

  {

    var template (value) NB_RadioBearerList_Type v_RadioBearerList;

    var integer i;

    var integer k;

    v_RadioBearerList[0] := cs_NB_SRB_Release(tsc_SRB1bis);

    if (p_IOT_State == USER_PLANE) {

      v_RadioBearerList[1] := cs_NB_SRB_Release(tsc_SRB1);
      i := lengthof(v_RadioBearerList);

      for (k:=0; k < lengthof(p_DrbList); k:= k+1) {

          v_RadioBearerList[i+k] := cs_NB_DRB_Release(p_DrbList[k]);

        }

    }

    f_NBIOT_SS_CommonRadioBearerConfig(p_CellId, v_RadioBearerList, p_TimingInfo, p_CnfFlag);

  }



After Change:

	function f_NBIOT_SS_SRBs_DRBs_Release(NBIOT_CellId_Type p_CellId,

                                        IOT_STATE_Type p_IOT_State,

                                        template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                        DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def,

                                        template (omit) boolean p_CnfFlag := omit,

                                        boolean p_DrbConfigured := true /*WA#*/) runs on NBIOT_PTC

  {

    var template (value) NB_RadioBearerList_Type v_RadioBearerList;

    var integer i;

    var integer k;

    v_RadioBearerList[0] := cs_NB_SRB_Release(tsc_SRB1bis);

    if (p_IOT_State == USER_PLANE) {

      v_RadioBearerList[1] := cs_NB_SRB_Release(tsc_SRB1);

      if (p_DrbConfigured) {
          i := lengthof(v_RadioBearerList);

          for (k:=0; k < lengthof(p_DrbList); k:= k+1) {

            v_RadioBearerList[i+k] := cs_NB_DRB_Release(p_DrbList[k]);

          }

      }
    }

    f_NBIOT_SS_CommonRadioBearerConfig(p_CellId, v_RadioBearerList, p_TimingInfo, p_CnfFlag);

  }



